```
>
```

Here's some Maple code used to solve the Model problem:

$$\varepsilon\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right)+\left(\frac{d}{dx}\mathrm{y}(x)\right)+\mathrm{y}(x)=0$$

y(0)=4

y(1)=5

```
> ODE:=epsilon*diff(y(x),x$2)+diff(y(x),x)+y(x);
```

$$ODE := \varepsilon\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right)+\left(\frac{d}{dx}\mathrm{y}(x)\right)+\mathrm{y}(x)$$

Here's the definition of things I described in class showing that it is indeed the exact solution:

```
> lp:=(-1+sqrt(1-4*epsilon))/(2*epsilon):
> lm:=(-1-sqrt(1-4*epsilon))/(2*epsilon):
> cp:=-(4*exp(lm)-5)/(exp(lp)-exp(lm)):
> cm:=(4*exp(lp)-5)/(exp(lp)-exp(lm)):
> Y:=cp*exp(lp*x)+cm*exp(lm*x);
```

$$Y := -\frac{\left(4\,\mathbf{e}^{\left(\frac{-1-\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}-5\right)\mathbf{e}^{\left(\frac{(-1+\sqrt{1-4\varepsilon})x}{2\varepsilon}\right)}}{\mathbf{e}^{\left(\frac{-1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}-\mathbf{e}^{\left(\frac{-1-\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}}+\frac{\left(4\,\mathbf{e}^{\left(\frac{-1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}-5\right)\mathbf{e}^{\left(\frac{(-1-\sqrt{1-4\varepsilon})x}{2\varepsilon}\right)}}{\mathbf{e}^{\left(\frac{-1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}-\mathbf{e}^{\left(\frac{-1-\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}}$$

```
> SATISFIES_ODE:=simplify(subs(y(x)=Y,ODE));
```

$$SATISFIES\_ODE := 0$$

```
> SATISFIES_BC_1:=simplify(subs(x=0,Y));
```

$$SATISFIES\_BC\_1 := 4$$

```
> SATISFIES_BC_2:=simplify(subs(x=1,Y));
```

$$SATISFIES\_BC\_2 := 5$$

Plot the solution for some different epsilon values

```
> Ye:=eps->subs(epsilon=eps,Y);
```

$$Ye := eps \rightarrow \mathrm{subs}(\varepsilon = eps, Y)$$

```
> plot({Ye(0.01),Ye(0.02),Ye(0.03)},x=0..1);
```




```
> series(lp,epsilon,3);
```

$$-1-\varepsilon+O(\varepsilon^2)$$

```
> series(lm,epsilon,1);
```

$$-\varepsilon^{-1}+1+O(\varepsilon)$$

```
> limit(cp,epsilon=0,right);
```

$$5\,\mathbf{e}$$

```
> limit(cm,epsilon=0,right);
```

$$4-5\,\mathbf{e}$$

Derivation of solution of BVP

```
> dsolve({ODE=0},y(x));
```

$$\{y(x) = \_C1\, e^{\left(\frac{(-1+\sqrt{1-4\varepsilon})x}{2\varepsilon}\right)} + \_C2\, e^{\left(-\frac{(1+\sqrt{1-4\varepsilon})x}{2\varepsilon}\right)}\}$$

```
> YE:=rhs(dsolve({ODE=0,y(0)=4,y(1)=5},y(x)));
```

$$YE := -\frac{\left(4\, e^{\left(-\frac{1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)} - 5\right) e^{\left(\frac{(-1+\sqrt{1-4\varepsilon})x}{2\varepsilon}\right)}}{e^{\left(\frac{-1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)} - e^{\left(-\frac{1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}} + \frac{\left(4\, e^{\left(\frac{-1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)} - 5\right) e^{\left(-\frac{(1+\sqrt{1-4\varepsilon})x}{2\varepsilon}\right)}}{e^{\left(\frac{-1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)} - e^{\left(-\frac{1+\sqrt{1-4\varepsilon}}{2\varepsilon}\right)}}$$

```
> Charpoly:=simplify(eval(subs(y(x)=exp(lambda*x),ODE))/exp(lambda*x
  ));
```

$$Charpoly := \varepsilon \lambda^2 + \lambda + 1$$

```
> solve(Charpoly=0,lambda);
```

$$\frac{-1+\sqrt{1-4\varepsilon}}{2\varepsilon}, -\frac{1+\sqrt{1-4\varepsilon}}{2\varepsilon}$$

```
> YG:=a*exp(lambda[p]*x)+b*exp(lambda[m]*x);
```

$$YG := a\, e^{(\lambda_p x)} + b\, e^{(\lambda_m x)}$$

```
> BC_Left:=simplify(subs(x=0,YG))=4;
```

$$BC\_Left := a + b = 4$$

```
> BC_Right:=simplify(subs(x=1,YG))=5;
```

$$BC\_Right := a\, e^{\lambda_p} + b\, e^{\lambda_m} = 5$$

```
> solve({BC_Left,BC_Right},{a,b});
```

$$\left\{b = -\frac{4\, e^{\lambda_p} - 5}{-e^{\lambda_p} + e^{\lambda_m}}, a = \frac{4\, e^{\lambda_m} - 5}{-e^{\lambda_p} + e^{\lambda_m}}\right\}$$

```
>
```